HW #4

1. Classical Uncertainty Principle

(a) The Maxwell's equations in vacuum are given by

$$\vec{\nabla} \cdot \vec{E} = 0$$
$$\vec{\nabla} \cdot \vec{B} = 0$$
$$\vec{\nabla} \times \vec{E} + \partial_t \vec{B} = 0$$
$$c^2 \vec{\nabla} \times \vec{B} - \partial_t \vec{E} = 0$$

In this problem, there are only x and t dependence, and the only non−vanishing components are E_y and B_z. Then the Maxwell's equations reduce to

$$\nabla_x E_y + \partial_t B_z = 0$$
$$-c^2 \nabla_x B_z - \partial_t E_y = 0$$

Putting them together, they reduce to a simple one−dimensional equation,

$$c^2 \nabla_x{}^2 E_y - \partial_t{}^2 E_y = 0.$$

Any function of the combination $c\,t - x$ satisfies this equation, namely

$$\left(c^2 \nabla_x{}^2 - \partial_t{}^2\right) f(c\,t - x) = 0.$$

Because the form of E_y given in the problem is a function of $c\,t - x$ only, it solves the Maxwell's equations automatically.

To simplify the problem, define $\gamma = 4\pi^2 \nu^2 \sigma^2 / c^2$. Then the electric field is:

```
Ey[x_, t_] = E0 * Sin[c Sqrt[γ]/σ t - Sqrt[γ]/σ x] E^-(x-c t)^2/(2 σ^2)
```

$$e^{-\frac{(-c\,t+x)^2}{2\sigma^2}}\ \mathrm{E0}\ \mathrm{Sin}\left[\frac{c\,t\sqrt{\gamma}}{\sigma} - \frac{x\sqrt{\gamma}}{\sigma}\right]$$

The form can be sketched as

```
Plot[Ey[x, t] /. {t → 0, γ → 400 π^2, E0 → 1, σ → 10}, {x, -30, 30}, PlotRange → {-1, 1}];
```



It oscillates just like the plane waves, but is localized. The "uncertainty" is defined using the formula analogous to the quantum mechanical wave function. First the "norm,"

```
norm = Integrate[Ey[x, 0]^2, {x, -∞, ∞}, Assumptions → {γ > 0, σ > 0}]
```

$$\frac{1}{2} e^{-\gamma} (-1 + e^{\gamma}) \, E0^2 \sqrt{\pi} \, \sigma$$

We could set the overall normalization $E_0 = 1$ throughout since it will drop out after taking the norm correctly into account. But we can also leave it in as a check that everything is working correctly.

Next the expectation value

```
Integrate[x * Ey[x, 0]^2, {x, -∞, ∞}, Assumptions → {γ > 0, σ > 0}]
```

$$0$$

OK, this vanishes. Finally the variance,

```
temp = Integrate[x^2 * Ey[x, 0]^2, {x, -∞, ∞}, Assumptions → {γ > 0, σ > 0}]
```

$$\frac{1}{4} e^{-\gamma} \, E0^2 \sqrt{\pi} \, (-1 + e^{\gamma} + 2\gamma) \, \sigma^3$$

```
variance = temp / norm
```

$$\frac{(-1 + e^{\gamma} + 2\gamma) \, \sigma^2}{2 (-1 + e^{\gamma})}$$

Note that the E0 dependence did in fact drop out.

```
FullSimplify[variance]
```

$$\frac{1}{2} \left(1 + \frac{2\gamma}{-1 + e^{\gamma}}\right) \sigma^2$$

One can write it as $(\Delta x)^2 = \frac{1}{2} \sigma^2 \left(1 + \frac{2\gamma}{e^{\gamma} - 1}\right)$, where $\gamma = 4 \pi^2 \nu^2 \sigma^2 / c^2$. It is especially simple when $\gamma \gg 1$, when $(\Delta x)^2 = \frac{1}{2} \sigma^2$.

(b) The Fourier transform to the frequency domain as a function of the variable "f" is calculated below. We might as well pick a specific position like "x=0" to evaluate the Fourier transform.

```
ft[f_] =
 Integrate[Ey[0, t] * E^(I 2 π f t), {t, -∞, ∞}, Assumptions → {γ > 0, σ > 0, c > 0, f ∈ Reals}]
```

Integrate::gener : Unable to check convergence. More…

$$\frac{i\, e^{-\frac{(c\sqrt{\gamma}+2 f \pi \sigma)^2}{2c^2}} \left(-1 + e^{\frac{4 f \pi \sqrt{\gamma}\, \sigma}{c}}\right) E0 \sqrt{\frac{\pi}{2}}\, \sigma}{c}$$

Again starting with the norm (noting that $\mathrm{ft}[f]^* \ \mathrm{ft}[f] = -\mathrm{ft}[f]^\wedge 2$):

```
norm2 = Integrate[-ft[f]^2, {f, 0, ∞}, Assumptions → {γ > 0, σ > 0, c > 0}]
```

$$\frac{e^{-\gamma} \left(-1 + e^{\gamma}\right) E0^2 \sqrt{\pi}\, \sigma}{4 c}$$

```
Integrate[-ft[f]^2, {f, -∞, ∞}, Assumptions → {γ > 0, σ > 0, c > 0}]
```

$$\frac{e^{-\gamma} \left(-1 + e^{\gamma}\right) E0^2 \sqrt{\pi}\, \sigma}{2 c}$$

Next, the average frequency,

```
exptf = Integrate[f * -ft[f]^2, {f, 0, ∞}, Assumptions → {γ > 0, σ > 0, c > 0}] / norm2
```

$$\frac{c\, e^{\gamma} \sqrt{\gamma}\, \mathrm{Erf}\left[\sqrt{\gamma}\right]}{2 \left(-1 + e^{\gamma}\right) \pi \sigma}$$

```
exptf /. σ → c √γ / (2 π ν)
```

$$\frac{e^{\gamma}\, \nu\, \mathrm{Erf}\left[\sqrt{\gamma}\right]}{-1 + e^{\gamma}}$$

```
Limit[Erf[√γ], γ → ∞]
```

1

One can write it as $\nu \frac{\mathrm{Erf}[\gamma^{1/2}]}{1-E^{-\gamma}}$, where $\gamma = 4\pi^2\ \nu^2\ \sigma^2 / c^2$. It is especially simple when $\gamma \gg 1$, when it reduces to nothing but ν. Finally the variance in the frequency is (remember to normalize!)

```
exptf2 = Integrate[f^2 * -ft[f]^2, {f, 0, ∞}, Assumptions → {γ > 0, σ > 0, c > 0}] / norm2
```

$$\frac{c^2 \left(-1 + e^{\gamma} (1 + 2\gamma)\right)}{8 \left(-1 + e^{\gamma}\right) \pi^2 \sigma^2}$$

```
FullSimplify[%]
```

$$\frac{c^2 \left(-1 + e^{\gamma} (1 + 2\gamma)\right)}{8 \left(-1 + e^{\gamma}\right) \pi^2 \sigma^2}$$

```
% /. σ → c √γ / (2 π ν) // Simplify
```

$$\frac{(-1+e^{\gamma}\,(1+2\,\gamma))\,\nu^2}{2\,(-1+e^{\gamma})\,\gamma}$$

```
Limit[%, γ → ∞]
```

ν^2

Then the square of the dispersion in the frequency is:

```
dispersion2 = exptf2 - exptf^2 // Simplify
```

$$-\frac{c^2\left(-(-1+e^{\gamma})\,(-1+e^{\gamma}\,(1+2\,\gamma))+2\,e^{2\,\gamma}\,\gamma\,\mathrm{Erf}\left[\sqrt{\gamma}\right]^2\right)}{8\,(-1+e^{\gamma})^2\,\pi^2\,\sigma^2}$$

```
dispersion2 /. σ → c √γ / (2 π ν) // Simplify
```

$$-\frac{\nu^2\left(-(-1+e^{\gamma})\,(-1+e^{\gamma}\,(1+2\,\gamma))+2\,e^{2\,\gamma}\,\gamma\,\mathrm{Erf}\left[\sqrt{\gamma}\right]^2\right)}{2\,(-1+e^{\gamma})^2\,\gamma}$$

```
Limit[dispersion2, γ → ∞]
% /. σ → c √γ / (2 π ν)
```

$$\frac{c^2}{8\,\pi^2\,\sigma^2}$$

$$\frac{\nu^2}{2\,\gamma}$$

Namely,

$$(\Delta\, f)^2 = \frac{\nu^2\left((1-e^{-\gamma})\,((1-e^{-\gamma})+2\,\gamma)-2\,\gamma\,\mathrm{Erf}[\gamma^{1/2}]^2\right)}{2\,\gamma\,(1-e^{-\gamma})^2}$$

which simplifies to

$$(\Delta\, f)^2 = \nu^2\,\frac{1}{2\,\gamma} = \frac{c^2}{8\,\pi^2\,\sigma^2}$$

when $\gamma \gg 1$. Therefore,

$$(\Delta\, x)^2\,(\Delta\, f)^2 = \frac{c^2}{16\,\pi^2}$$

Once interpreted as a photon,

$$(\Delta\, f)^2 = c^2\,(\Delta\, p)^2 \big/ h^2$$

, and hence

$$(\Delta x)^2 \, (\Delta p)^2 = \frac{\hbar^2}{4}$$

as expected.

2. Gaussian Wave Packet

(a) The norm is

$$\langle \psi \mid \psi \rangle = \int \langle \psi \mid \mathrm{x} \rangle \langle \mathrm{x} \mid \psi \rangle \, d\mathrm{x} = \int [\psi(\mathrm{x})]^* \, \psi(\mathrm{x}) \, d\mathrm{x}$$

. Specify unnormalized state on spatial basis :

```
ψun[x_] = E^(I p x / ħ) E^(-(x-x0)^2/(4 d^2));
```

We now integrate:

```
norm3 =
 Integrate[ComplexExpand[Conjugate[ψun[x]] ψun[x]], {x, -∞, ∞}, Assumptions → {d > 0}]
```

$d \sqrt{2\pi}$

Therefore N is

```
1 / Sqrt[norm3]
```

$$\frac{1}{\sqrt{d} \, (2\pi)^{1/4}}$$

Verify:

```
ψ[x_] = ψun[x] / Sqrt[norm3];

Integrate[ComplexExpand[Conjugate[ψ[x]] ψ[x]], {x, -∞, ∞}, Assumptions -> {d > 0}]
```

1

(b) Now calculate

$$\langle \mathrm{x} \rangle = \langle \psi \mid \mathrm{x} \mid \psi \rangle = \int \langle \psi \mid \mathrm{x} \rangle \, \mathrm{x} \, \langle \mathrm{x} \mid \psi \rangle \, d\mathrm{x} = \int [\psi(\mathrm{x})]^* \, \mathrm{x} \, \psi(\mathrm{x}) \, d\mathrm{x}$$

:

```
Integrate[ComplexExpand[Conjugate[ψ[x]] x ψ[x]], {x, -∞, ∞}, Assumptions -> {d > 0}]
```

x0

(c) Calculate 2nd moment

$$\langle x^2 \rangle = \langle \psi \mid \mathrm{x}^2 \mid \psi \rangle = \int \langle \psi \mid \mathrm{x} \rangle \, \mathrm{x}^2 \, \langle \mathrm{x} \mid \psi \rangle \, d\mathrm{x} = \int [\psi(\mathrm{x})]^* \, \mathrm{x}^2 \, \psi(\mathrm{x}) \, d\mathrm{x}$$

:

```
Integrate[ComplexExpand[Conjugate[ψ[x]] x^2 ψ[x]], {x, -∞, ∞}, Assumptions → {d > 0}]
```

$d^2 + x0^2$

So our dispersion $\Delta x = \sqrt{\langle x^2 \rangle - \langle x \rangle^2}$ is

```
dispx = Simplify[Sqrt[% - x0^2], Assumptions → {d > 0}]
```

d

(d) We utilize the unity operator, noting that the momentum eigenvalue p in ψ(x) is not the same as the momentum coordinate q in ϕ(q),

$$\phi(q) = \langle q|\psi\rangle = \int \langle q \mid x\rangle \langle x \mid \psi\rangle \, dx = \int \frac{e^{-i q x/\hbar}}{\sqrt{2\pi\hbar}} \psi(x) \, dx$$

:

```
ϕ[q_] = TrigToExp[Simplify[
   ComplexExpand[Integrate[Exp[-I q x / ℏ] / Sqrt[2 π ℏ] ψ[x], {x, -∞, ∞}, Assumptions → {d > 0, ℏ > 0}]],
   Assumptions → {d > 0, ℏ > 0}]]
```

Integrate::gener : Unable to check convergence. More…

$$\frac{d \, e^{-\frac{d^2 (p-q)^2}{\hbar^2} + \frac{i (p-q) x0}{\hbar}} \left(\frac{2}{\pi}\right)^{1/4}}{\sqrt{\hbar \, d}}$$

Check norm:

```
Integrate[ComplexExpand[Conjugate[ϕ[q]] ϕ[q]], {q, -∞, ∞}, Assumptions → {d > 0, ℏ > 0}]
```

1

(e) Similar to $\langle x \rangle$:

```
Integrate[ComplexExpand[Conjugate[ϕ[q]] q ϕ[q]], {q, -∞, ∞}, Assumptions → {d > 0, ℏ > 0}]
```

p

(f) Let us now calculate Δp similarly to Δx:

```
Integrate[ComplexExpand[Conjugate[ϕ[q]] q^2 ϕ[q]], {q, -∞, ∞}, Assumptions → {d > 0, ℏ > 0}]
```

$$\frac{\hbar^2}{4 d^2} + p^2$$

```
dispp = Simplify[Sqrt[% - p^2], Assumptions → {d > 0, ℏ > 0}]
```

$$\frac{\hbar}{2\,d}$$

Finally, compute the combined uncertainty ΔxΔp :

```
dispx * dispp
```

$$\frac{\hbar}{2}$$

We obtain the minimum combined uncertainty as expected!

3. Proving the Baker−Hausdorff Lemma

(a) $\frac{d}{dt} B(t) = e^{tA}\, A\, B\, e^{-tA} + e^{tA}\, B\,(-A)\, e^{-tA}$

$$= e^{tA}\,[A,\ B]\, e^{-tA}$$

Noting that:

$$e^{tA}\,[A,\ B]\, e^{-tA} = [A,\ B(t)]$$

(b) $B(1) = B + \int_0^1 \frac{d}{dt} B(t)\, dt$

$$B(1) = B + \int_0^1 [A,\ B(t)]\, dt$$

$$\sum_{n=0}^{\infty} B_n = B_0 + \sum_{n=0}^{\infty} [A, B_n] \int_0^1 t^n\, dt$$

$$\sum_{n=0}^{\infty} B_n = B_0 + \sum_{n=0}^{\infty} [A, B_n]\, \frac{1}{n+1}$$

$$\sum_{n=1}^{\infty} B_n = \sum_{n=0}^{\infty} [A, B_n]\, \frac{1}{n+1}$$

$$\sum_{n=1}^{\infty} B_n = \sum_{n=1}^{\infty} [A, B_{n-1}]\, \frac{1}{n}$$

$$B_n = \frac{1}{n}[A, B_{n-1}]$$

(c) $B_n = \frac{1}{n}[A, B_{n-1}] = \frac{1}{n}\left[A,\ \frac{1}{n-1}[A,\ B_{n-2}]\right] = \frac{1}{n(n-1)}[A,\ [A,\ B_{n-2}]] = \ldots$

$$= \frac{1}{n(n-1)\ldots(n-n+1)}[A,\ [A,\ \ldots\ [A,\ B_{n-n}]\ \ldots]]$$

$$B_n = \frac{1}{n!}[A,\ [A,\ \ldots\ [A,\ B]\ \ldots]] \qquad \text{(w/ } n \text{ nested commutators)}$$

(d) $e^{A}\, B\, e^{-A} = B(1) = \sum_{n=0}^{\infty} B_n$

$$e^A B e^{-A} = \sum_{n=0}^{\infty} \frac{1}{n!}[A, [A, \ldots [A, B] \ldots]]$$

(e) $e^{i p a/\hbar} x e^{-i p a/\hbar} = \sum_{n=0}^{\infty} \frac{1}{n!}[A, [A, \ldots [A, x] \ldots]]$

$$= \sum_{n=0}^{\infty} \frac{a^n}{n!} \left(\frac{i}{\hbar}\right)^n [p, [p, \ldots [p, x] \ldots]]$$

Only the first and second terms survive:

$$= x + a\left(\frac{i}{\hbar}\right)(-i\hbar)$$

$$e^{i p a/\hbar} x e^{-i p a/\hbar} = x + a$$